

Logisch Programmeren en Zoektechnieken

Ulle Endriss

Institute for Logic, Language and Computation

University of Amsterdam

[<http://www.illc.uva.nl/~ulle/teaching/prolog/>]

What is Prolog?

- Prolog (*programming in logic*) is a logic-based programming language: programs correspond to sets of logical formulas and the Prolog interpreter uses logical methods to resolve queries.
- Prolog is a *declarative* language: you specify *what* problem you want to solve rather than *how* to solve it.
- Prolog is very useful in *some* problem areas, such as artificial intelligence, natural language processing, databases, . . . , but pretty useless in others, such as for instance graphics or numerical algorithms.
- The objective of this first lecture is to introduce you to the most basic concepts of the Prolog programming language.

Facts

A little Prolog program consisting of four *facts*:

```
bigger(elephant, horse).
```

```
bigger(horse, donkey).
```

```
bigger(donkey, dog).
```

```
bigger(donkey, monkey).
```

Queries

After compilation we can *query* the Prolog system:

```
?- bigger(donkey, dog).
```

Yes

```
?- bigger(monkey, elephant).
```

No

A Problem

The following query does not succeed!

```
?- bigger(elephant, monkey).
```

No

The *predicate* `bigger/2` apparently is not quite what we want.

What we'd really like is what is known as the *transitive closure* of `bigger/2`. In other words: a predicate that succeeds whenever it is possible to go from the first animal to the second by iterating the previously defined facts.

Rules

The following two *rules* define `is_bigger/2` as the transitive closure of `bigger/2` (via *recursion*):

`is_bigger(X, Y) :- bigger(X, Y).`

`is_bigger(X, Y) :- bigger(X, Z), is_bigger(Z, Y).`

↑	↑
“if”	“and”

Now it works

```
?- is_bigger(elephant, monkey).
```

Yes

Even better, we can use the *variable* X:

```
?- is_bigger(X, donkey).
```

```
X = horse ;
```

```
X = elephant ;
```

No

Press ; (semicolon) to find alternative solutions. No at the end indicates that there are no further solutions.

Another Example

Are there any animals that are both smaller than a donkey and bigger than a monkey?

```
?- is_bigger(donkey, X), is_bigger(X, monkey).
```

No

Next: basic terminology for talking about Prolog ...

Terms

Prolog *terms* are either

- *numbers*,
- *atoms*,
- *variables*, or
- *compound terms*.

Atoms start with a lowercase letter or are enclosed in single quotes:

elephant, xYZ, a_123, 'Another pint please'

Variables start with a capital letter or the underscore:

X, Elephant, _G177, MyVariable, _

Terms (continued)

Compound terms have a *functor* (an atom) and a number of *arguments* (terms):

`is_bigger(horse, X)`

`f(g(Alpha, _), 7)`

`'My Functor'(dog)`

Atoms and numbers are called *atomic terms*.

Atoms and compound terms are called *predicates*.

Terms without variables are called *ground terms*.

Facts and Rules

Facts are predicates followed by a full stop. Facts are used to define something as being unconditionally true.

```
bigger(elephant, horse).  
parent(john, mary).
```

Rules consist of a *head* and a *body* separated by `:-`. The head of a rule is true if all predicates in the body can be proved to be true.

```
grandfather(X, Y) :-  
    father(X, Z),  
    parent(Z, Y).
```

Programs and Queries

Programs: Facts and rules are called *clauses*. A Prolog program is a list of clauses.

Queries are predicates (or sequences of predicates) followed by a full stop. They are typed in at the Prolog prompt and cause the system to reply.

```
?- is_bigger(horse, X), is_bigger(X, dog).
```

```
X = donkey
```

```
Yes
```

Built-in Predicates

- Compiling a program file:

```
?- consult('big-animals.pl').
```

Yes

- Writing terms on the screen:

```
?- write('Hello World!'), nl.
```

Hello World!

Yes

Next: how Prolog actually works ...

Matching

Two terms *match* if they are either identical or if they can be made identical by substituting their variables with suitable ground terms.

We can explicitly ask Prolog whether two given terms match by using the equality-predicate `=` (written as an infix operator).

```
?- born(mary, yorkshire) = born(mary, X).
```

```
X = yorkshire
```

```
Yes
```

The variable instantiations are reported in Prolog's answer.

Examples

$?- f(a, g(X, Y)) = f(X, Z), Z = g(W, h(X)).$

$X = a$

$Y = h(a)$

$Z = g(a, h(a))$

$W = a$

Yes

$?- p(X, 2, 2) = p(1, Y, X).$

No

The Anonymous Variable

The variable `_` (underscore) is called the *anonymous variable*.

Every occurrence of `_` represents a different variable (which is why instantiations are not being reported).

```
?- p(_, 2, 2) = p(1, Y, _).
```

```
Y = 2
```

```
Yes
```


Answering Queries

Answering a query means proving that the goal represented by that query can be satisfied (according to the programs that are currently in memory).

Recall: Programs are lists of facts and rules. A fact declares something as being true. A rule states conditions for a statement being true.

Answering Queries (continued)

- If a goal matches with a *fact*, then it is satisfied.
- If a goal matches the *head of a rule*, then it is satisfied if the goal represented by the rule's body is satisfied.
- If a goal consists of several *subgoals* separated by commas, then it is satisfied if all its subgoals are satisfied.
- When trying to satisfy goals with built-in predicates such as `write/1`, Prolog also performs the associated action (e.g., writing something on the screen).

Example: Mortal Philosophers

Consider the following argument:

All men are mortal.

Socrates is a man.

Hence, Socrates is mortal.

It has two *premises* and a *conclusion*.

Translating it into Prolog

The two premises can be expressed as a little Prolog program:

```
mortal(X) :- man(X).  
man(socrates).
```

The conclusion can then be formulated as a query:

```
?- mortal(socrates).  
Yes
```

Goal Execution

- (1) The query `mortal(socrates)` is made the initial goal.
- (2) Prolog looks for the first matching fact or head of rule and finds `mortal(X)`. Variable instantiation: `X = socrates`.
- (3) This variable instantiation is extended to the rule's body, i.e., `man(X)` becomes `man(socrates)`.
- (4) New goal: `man(socrates)`.
- (5) Success, because `man(socrates)` is a fact itself.
- (6) Therefore, also the initial goal succeeds.

Programming Style

It is extremely important that you write programs that are easily understood by others! Some guidelines:

- Use *comments* to explain what you are doing:

```
/* This is a long comment, stretching over several  
lines, which explains in detail how I have implemented  
the aunt/2 predicate ... */
```

```
aunt(X, Z) :-  
    sister(X, Y), % This is a short comment.  
    parent(Y, Z).
```

Programming Style (continued)

- Separate clauses by one or more blank lines.
- Write only one predicate per line and use indentation:

```
blond(X) :-  
    father(Father, X),  
    blond(Father),  
    mother(Mother, X),  
    blond(Mother).
```

(Very short clauses may also be written in a single line.)

- Insert a space after every comma inside a compound term:

```
born(mary, yorkshire, '01/01/1995')
```

- Write short clauses with bodies consisting of only a few goals.
If necessary, split into shorter sub-clauses.
- Choose meaningful names for your variables and atoms.

Summary: Terminology

- All Prolog expressions are made up of *terms* (numbers, atoms, variables, or compound terms).
- *Atoms* start with lowercase letters or are enclosed in single quotes; *variables* start with capital letters or the underscore.
- Prolog programs are lists of *facts* and *rules* (*clauses*).
- *Queries* are submitted to the system to initiate a computation.
- Some *built-in predicates* have special meaning.

Summary: Answering Queries

- When answering a query, Prolog tries to prove that the corresponding goal is satisfiable (can be made true). This is done using the rules and facts given in a program.
- A goal is executed by *matching* it with the first possible fact or head of a rule. In the latter case, the rule's body becomes the new goal.
- The variable instantiations made during matching are carried along throughout the computation and reported at the end.
- Only the *anonymous variable* `_` can be instantiated differently whenever it occurs.